
Abstract. Alperin’s Weight Conjecture was originally formulated for alge-
braically closed fields (see cite [1]). For some families of groups – such as the
symmetric groups – it is known to hold for arbitrary fields (see cite [2]), so it
is reasonable to ask whether this conjecture holds for arbitrary fields, and in
particular, if it holds for finite fields. We wrote computer software in MAGMA
(see cite [8]) to test Alperin’s Weight Conjecture for finite fields, and tested this
software on groups of small order and the prime fields whose characteristics di-
vide the order of the groups. We found no counterexamples to this version of
Alperin’s Conjecture for groups of order up to 100.

Resumen. La conjetura de pesos de Alperin fue formulada originalmente para
campos algebraicamente cerrados. Para algunas familias de grupos –como por
ejemplo los grupos simétricos– esta Conjetura es válida para todos los cam-
pos, y en particular, para los campos finitos. Es razonable preguntar si dicha
Conjetura permanecerá válida para todos los grupos y todos los campos, y
en particular para los campos finitos. En este art́ıculo verificamos (usando
MAGMA) la Conjetura de Pesos de Alperin para todos los grupos de orden
menor o igual a 100 y todos los campos primos cuyas caracteŕısticas dividen el
orden de cada grupo.

Palabras y frases clave. Grupo de representaciones, conjetura de Alperin, peso,
software, computacional.

1. Introduction

In the modern representation theory of groups, Alperin’s Weight Conjecture
remains as one of the most important and difficult open problems. This con-
jecture was originally formulated for algebraically closed fields of prime char-
acteristic, as it appears on [1]. However, for some families of groups (such
as the symmetric groups) Alperin’s Conjecture is known to hold for arbitrary
fields (see [2]). A natural question to ask is whether the conjecture will hold
in general for all finite groups and all fields. The first step towards answering
this question is to consider the case of the finite fields, particularly the prime
fields. In other words, is it true that, given an arbitrary finite group, Alperin’s
Conjecture holds for all the prime fields whose characteristics divide the order
of that group? We approached this problem from a computacional perspective.

We wrote computer software in Magma (see [8]) to test Alperin’s Conjecture
for finite fields. We ran our software on all the groups in Magma’s library of
small groups up to order 100, and on the appropriate prime fields. We found
no counterexamples to this finite version of Alperin’s Conjecture.

In Section 2 we define the basic concepts and formulate Alperin’s Conjecture.
In Section 3 we describe the software that we wrote, and we present the results
that we obtained using our software.

2. Alperin’s Conjecture

We give the definition of weight and formulate Alperin’s Conjecture in its most
general form. We also mention some classes of groups for which it is known to
be valid.

Throughout this section, G will be a finite group, p a prime number, and
k an algebraically closed field of characteristic p, unless otherwise stated. All
our modules will be finite dimensional over k.

Definition 2.1. A weight for kG is a pair (Q, S) where Q is a p-subgroup
and S is a simple module for k[NG(Q)] which is projective when regarded as a
module for k[NG(Q)/Q]. We sometimes refer to S as a weight module, and Q
as its weight subgroup.

Remark 2.2. Since S is k[N(Q)]-simple and Q is a p-subgroup of NG(Q), it
follows that Q acts trivially on S, so S is also a k[NG(Q)/Q]-module and the
definition makes sense. Moreover, S is k[NG(Q)/Q]-simple as well.

Remark 2.3. If we replace S by an isomorphic k[NG(Q)]-module we consider
this the same weight, and we make the same identification when we replace Q
by a conjugate subgroup (so that the normalizers will be conjugate, too).

Now we can formulate the main problem that we shall discuss in this section.

Volumen 41, Número 2, Año 2007



FINITE VERSION OF ALPERIN’S CONJECTURE 327

Conjecture 2.3.1 (Alperin’s Conjecture). The number of weights for kG
equals the number of simple kG-modules.

A stronger version of the preceding statement is that there is a bijection
within each block of the group algebra.

Definition 2.4. If (Q, S) is a weight for kG, then S belongs to a block b of
NG(Q) and this block corresponds with a block B of G via the Brauer corre-
spondence; hence we can say that the weight (Q, S) belongs to the block B of
kG so the weights are partitioned into blocks.

Conjecture 2.4.1 (Alperin’s Conjecture, Block Form). The number of weights
in a block of kG equals the number of simple modules in the block.

This version of the conjecture implies the original one, as it can be obtained
by summing the equalities from the stronger conjecture over the blocks. This
stronger conjecture has been proved when G is a:

• Finite group of Lie type and characteristic p (Cabanes, [9]).
• Soluble group (Okuyama).
• Symmetric group (Alperin and Fong, [2]).
• GL(n, q), p odd and p does not divide q (Alperin and Fong, [2]).
• GL(n, q), p = 2 and q odd (An, [3]).

The conjecture has also been checked in a variety of other cases (see [4], [5], [6], [7]).

3. Software

Here is the software that we wrote to test Alperin’s Conjecture experimentally.

/* Magma routines to test Alperin’s Conjecture on finite fields.

Let G be a finite group, F a field.

/* ------------------------------------------------------------- */

number_of_simple_projective_modules := function(G,F)

/* Return the number of simple projective modules for FG. */

triv := sub< G | [Identity(G)] >;

regular_mod := PermutationModule(G,triv,F);

proj_mod := IndecomposableSummands(regular_mod);

answer := 0;

irrmod := IrreducibleModules(G, F);

for m in irrmod do

for pj in proj_mod do

if Dimension(m) eq Dimension(pj) then

if IsIsomorphic(m,pj) then

answer +:= 1;

break;

end if;

end if;

end for;

end for;

return answer;

end function;

highest_power := function(p,n)

/* Return the highest power of p dividing n. */

answer := 1;

bool, qqu := IsDivisibleBy(n,p);

while bool do

answer *:= p;

bool, qqu := IsDivisibleBy(qqu,p);

end while;

return answer;

end function;

number_of_weights := function(G,F)

/* Return the number of weights for FG. */

p := Characteristic(F);

pn := highest_power(p,Order(G));

subcc := SubgroupClasses(G : OrderDividing := pn);

answer := 0;

for subb in subcc do

qgp := Normalizer(G,subb‘subgroup) / subb‘subgroup ;

answer +:= number_of_simple_projective_modules( qgp, F);

end for;

return answer;

end function;

how_many_irreducibles := function(G,F)

return #IrreducibleModules(G,F);

end function;

test_conjecture_group := function(G,F)

if IsNormal(G,SylowSubgroup(G,Characteristic(F))) then

return true;

else

return (number_of_weights(G,F) eq how_many_irreducibles(G,F));

end if;

end function;

test_conjecture_order := function(n)

flag := true;

counterex := [**];

if IsPrimePower(n) then

return true, counterex;

end if;

lgrp := SmallGroups(n);

pprimes := Factorization(n);

ffields := [];

for pp in pprimes do

Append(~ffields, GF(pp[1]));

end for;

for G in lgrp do

if 1 eq 1 then

for ff in ffields do

if not(test_conjecture_group(G,ff)) then

Append(~counterex,<IdentifyGroup(G),G,ff>);

flag := false;

//return false;

end if;

end for;

end if;

end for;

if flag then

return true, counterex;

else

return false, counterex;

end if;

end function;

ttry := function(a,b)

counter := [**];

for n := a to b do

flag, info := test_conjecture_order(n);

if not(flag) then

Append(~counter, <n,info>);

end if;

end for;

if #counter eq 0 then

return true;

else

return counter;

end if;

end function;

Here is the data we gathered running our software on the library of small
groups of Magma. We examined all groups of order less than or equal to 100,
and for each of these groups, all the prime fields whose characteristics divide
the order of the group. It took a little over fifteen minutes to get these results.

> ttry(2,100);

true

From this we can see that there are no counterexamples to Alperin’s Weight
Conjecture for groups of order up to 100 and their prime fields.

4. Epilogue

We believe Alperin’s Conjecture to be true for all finite groups and all fields. We
intend to streamline our software to run it on more groups, including several
sporadic simple groups. This paper is the core of the MSc Thesis of Adán
Cortés-Medina, which was supervised by Luis Valero-Elizondo.

Acknowledgments. We are grateful to Alberto G. Raggi-Cárdenas for point-
ing out a simplification in our algorithm, namely, that Alperin’s Conjecture
is true for any field of characteristic p if the group G has only one Sylow p-
subgroup.

